

NAG Library Routine Document

F07FGF (DPOCON)

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of *bold italicised* terms and other implementation-dependent details.

1 Purpose

F07FGF (DPOCON) estimates the condition number of a real symmetric positive definite matrix A , where A has been factorized by F07FDF (DPOTRF).

2 Specification

```
SUBROUTINE F07FGF (UPLO, N, A, LDA, ANORM, RCOND, WORK, IWORK, INFO)
INTEGER          N, LDA, IWORK(N), INFO
REAL (KIND=nag_wp) A(LDA,*), ANORM, RCOND, WORK(3*N)
CHARACTER(1)    UPLO
```

The routine may be called by its LAPACK name *dpocon*.

3 Description

F07FGF (DPOCON) estimates the condition number (in the 1-norm) of a real symmetric positive definite matrix A :

$$\kappa_1(A) = \|A\|_1 \|A^{-1}\|_1.$$

Since A is symmetric, $\kappa_1(A) = \kappa_\infty(A) = \|A\|_\infty \|A^{-1}\|_\infty$.

Because $\kappa_1(A)$ is infinite if A is singular, the routine actually returns an estimate of the **reciprocal** of $\kappa_1(A)$.

The routine should be preceded by a call to F06RCF to compute $\|A\|_1$ and a call to F07FDF (DPOTRF) to compute the Cholesky factorization of A . The routine then uses Higham's implementation of Hager's method (see Higham (1988)) to estimate $\|A^{-1}\|_1$.

4 References

Higham N J (1988) FORTRAN codes for estimating the one-norm of a real or complex matrix, with applications to condition estimation *ACM Trans. Math. Software* **14** 381–396

5 Parameters

- 1: UPLO – CHARACTER(1) *Input*
On entry: specifies how A has been factorized.
 UPLO = 'U'
 $A = U^T U$, where U is upper triangular.
 UPLO = 'L'
 $A = L L^T$, where L is lower triangular.
Constraint: UPLO = 'U' or 'L'.
- 2: N – INTEGER *Input*
On entry: n , the order of the matrix A .
Constraint: $N \geq 0$.

- 3: A(LDA,*) – REAL (KIND=nag_wp) array Input
Note: the second dimension of the array A must be at least $\max(1, N)$.
On entry: the Cholesky factor of A, as returned by F07FDF (DPOTRF).
- 4: LDA – INTEGER Input
On entry: the first dimension of the array A as declared in the (sub)program from which F07FGF (DPOCON) is called.
Constraint: $LDA \geq \max(1, N)$.
- 5: ANORM – REAL (KIND=nag_wp) Input
On entry: the 1-norm of the **original** matrix A, which may be computed by calling F06RCF with its parameter NORM = '1'. ANORM must be computed either **before** calling F07FDF (DPOTRF) or else from a **copy** of the original matrix A.
Constraint: $ANORM \geq 0.0$.
- 6: RCOND – REAL (KIND=nag_wp) Output
On exit: an estimate of the reciprocal of the condition number of A. RCOND is set to zero if exact singularity is detected or the estimate underflows. If RCOND is less than **machine precision**, A is singular to working precision.
- 7: WORK(3 × N) – REAL (KIND=nag_wp) array Workspace
- 8: IWORK(N) – INTEGER array Workspace
- 9: INFO – INTEGER Output
On exit: INFO = 0 unless the routine detects an error (see Section 6).

6 Error Indicators and Warnings

INFO < 0

If INFO = -i, argument i had an illegal value. An explanatory message is output, and execution of the program is terminated.

7 Accuracy

The computed estimate RCOND is never less than the true value ρ , and in practice is nearly always less than 10ρ , although examples can be constructed where RCOND is much larger.

8 Parallelism and Performance

F07FGF (DPOCON) is not threaded by NAG in any implementation.

F07FGF (DPOCON) makes calls to BLAS and/or LAPACK routines, which may be threaded within the vendor library used by this implementation. Consult the documentation for the vendor library for further information.

Please consult the X06 Chapter Introduction for information on how to control and interrogate the OpenMP environment used within this routine. Please also consult the Users' Note for your implementation for any additional implementation-specific information.

9 Further Comments

A call to F07FGF (DPOCON) involves solving a number of systems of linear equations of the form $Ax = b$; the number is usually 4 or 5 and never more than 11. Each solution involves approximately $2n^2$ floating-point operations but takes considerably longer than a call to F07FEF (DPOTRS) with one right-hand side, because extra care is taken to avoid overflow when A is approximately singular.

The complex analogue of this routine is F07FUF (ZPOCON).

10 Example

This example estimates the condition number in the 1-norm (or ∞ -norm) of the matrix A , where

$$A = \begin{pmatrix} 4.16 & -3.12 & 0.56 & -0.10 \\ -3.12 & 5.03 & -0.83 & 1.18 \\ 0.56 & -0.83 & 0.76 & 0.34 \\ -0.10 & 1.18 & 0.34 & 1.18 \end{pmatrix}.$$

Here A is symmetric positive definite and must first be factorized by F07FDF (DPOTRF). The true condition number in the 1-norm is 97.32.

10.1 Program Text

```

Program f07fgfe

!      F07FGF Example Program Text

!      Mark 25 Release. NAG Copyright 2014.

!      .. Use Statements ..
      Use nag_library, Only: dlansy => f06rcf, dpocon, dpotrf, nag_wp, x02ajf
!      .. Implicit None Statement ..
      Implicit None
!      .. Parameters ..
      Integer, Parameter          :: nin = 5, nout = 6
!      .. Local Scalars ..
      Real (Kind=nag_wp)          :: anorm, rcond
      Integer                     :: i, info, lda, n
      Character (1)               :: uplo
!      .. Local Arrays ..
      Real (Kind=nag_wp), Allocatable :: a(:,,:), work(:)
      Integer, Allocatable         :: iwork(:)
!      .. Executable Statements ..
      Write (nout,*) 'F07FGF Example Program Results'
!      Skip heading in data file
      Read (nin,*)
      Read (nin,*) n
      lda = n
      Allocate (a(lda,n),work(3*n),iwork(n))

!      Read A from data file

      Read (nin,*) uplo
      If (uplo=='U') Then
         Read (nin,*)(a(i,i:n),i=1,n)
      Else If (uplo=='L') Then
         Read (nin,*)(a(i,1:i),i=1,n)
      End If

!      Compute norm of A
!      f06rcf is the NAG name equivalent of the LAPACK auxiliary dlansy
      anorm = dlansy('1-norm',uplo,n,a,lda,work)

!      Factorize A
!      The NAG name equivalent of dpotrf is f06fdf
      Call dpotrf(uplo,n,a,lda,info)

      Write (nout,*)

```

```

      If (info==0) Then
!       Estimate condition number
!
!       The NAG name equivalent of dpocon is f07fgf
      Call dpocon(uplo,n,a,lda,anorm,rcond,work,iwork,info)

      If (rcond>=x02ajf()) Then
        Write (nout,99999) 'Estimate of condition number =', &
          1.0_nag_wp/rcond
      Else
        Write (nout,*) 'A is singular to working precision'
      End If
    Else
      Write (nout,*) 'A is not positive definite'
    End If

99999 Format (1X,A,1P,E10.2)
      End Program f07fgfe

```

10.2 Program Data

```

F07FGF Example Program Data
  4                               :Value of N
  'L'                             :Value of UPLO
  4.16
 -3.12   5.03
  0.56  -0.83   0.76
 -0.10   1.18   0.34   1.18   :End of matrix A

```

10.3 Program Results

```

F07FGF Example Program Results

Estimate of condition number =  9.73E+01

```
